

# GERÇEK ZAMAN ENTEGRE DESTEKLİ PIC MİKROKONTROLÖR PROJESİ

**Prof. Dr. Doğan İbrahim**

*Yakın Doğu Üniversitesi, Mühendislik Fakültesi, Biyomedikal Mühendisliği, Bölümü*

*E-mail: dogan @neu.edu.tr Tel: 90 3922236464*

## 1. GİRİŞ

Bu yazımızda Philips firmasının üretmiş olduğu PCF8583 gerçek zaman entegresinin yapısını ve bu entegrenin PIC mikrokontrolör tabanlı sistemlerde kullanımını göreceğiz.

## 2. PCF8583 GERÇEK ZAMAN ENTEGRESİ


8 bacaklı olan bu entegre gerçek tarih ve zamana ilaveten üzerinde alarm devresi ve 256 bayt RAM bellek de bulunmaktadır. Bu yazımızda sadece gerçek zaman tarih ve saatin çalışmasını göreceğiz.

PCF8583<sup>1</sup> entegre devre I<sup>2</sup>C protokolünü kullanarak çalışmaktadır. Bu protokole göre entegre dış dünya ile iletişim kurmak için SDA ve SCL olarak isimlendirilen iki bacak kullanmaktadır. Bu bacaklar normal olarak 10K direnç ile +V güç kaynağına bağlanmaktadır. mikro BASIC derleyici<sup>2</sup> I<sup>2</sup>C protokolünü desteklemekte ve bu protokol için çeşitli fonksiyonları bulunmaktadır.

PCF8583 entegresi normal olarak +5V ile çalışmaktadır. Güç kaynağı olmayan durumlarda tarih ve saatin bozulmaması için entegreyi pil ile çalıştırmak mümkündür.

Şekil 1 de PCF8583 entegrenin bacak bağlantıları gösterilmiştir. 8 bacaklı olan bu entegrenin bacak tanımları şöyledir:

OSC1, OSC0: 32768Hz kristal bağlantısı  
A0: Entegre adres bacağı  
VSS: Toprak bağlantısı  
SDA, SCL: I<sup>2</sup>C bacakları  
INT: Kesme çıkış bacağı  
VDD: +V güç kaynağı bacağı



Şekil 1 PCF8583 bacak bağlantıları

Şekil 2 de görüleceği gibi PCF8583 entegresinin tarih ve saat fonksiyonlarını kontrol eden 8 tane yazmaç bulunmaktadır. Bu yazmaçlardan 0 nolu adresi olan yazmaç **kontrol/statüs** yazmacı olarak bilinmektedir. Bu yazmacın şu bitleri bulunmaktadır:

Bit 0: zamanlama bayrağı  
Bit 1: alarm bayrağı  
Bit 2: alarm aktifleme biti  
Bit 3: mask bayrağı  
Bit 4: Fonksiyon modu  
Bit 5: son sayım bayrağı  
Bit 6: sayımı durdurma bayrağı

kontrol/statüs	
1/100 saniye	
1/10s	1/100s
Saniye	
10s	1s
Dakika	
10dak	1dak
Saat	
10saat	1saat
yıl/tarih	
10gün	1gün
hafta günü/ay	
10ay	1ay
Zamanlayıcı	
10gün	1gün

Şekil 2 PCF8583 entegresinin tarih ve saat yazmaçları


PCF8583 tarih ve saat değerleri Tablo 1 de verilen değerleri almaktadırlar.

Tablo 1 PCF8583 tarih ve saat değerleri

1/100 saniye	00 – 99
Saniye	00 – 59
Dakika	00 – 59
saat (24h)	00 – 23
saat (12h)	01AM – 12PM
tarih	01 – 31
ay	01 – 12
Yıl	0 – 3
hafta günü	0 – 6
zamanlayıcı	00 - 99

### 3. PCF8583 GERÇEK ZAMAN ENTEGRESİNE ÖRNEK

Bu bölümde PCF8583 gerçek zaman entegresine bir örnek verilmiştir. Şekil 3 de bu örneğin blok şeması verilmiştir. Bu örnekte PCF8583 entegresi PIC18F452<sup>3</sup> mikrokontrolöre bağlanmıştır. Devrede güç kaynağı olmadığı zamanlarda tarih ve saat değerlerinin bozulmaması için PCF8583 entegresine bir pil bağlanmıştır. PCF8583 entegresi 32768 Hz kristal ile çalışmaktadır. örnekte PCF8583 entegresinin tarih ve saat değerleri ilk olarak RS232 seri port üzerinden ve bir PC tarafından yüklenmektedir. Devrede bulunan bir LCD de her saniye gerçek zaman şu formatta gösterilmektedir (aşağıdaki örnekte saatin 10 ve tarihin 12 Eylül olduğu örnek olarak alınmıştır):



Şekil 3 Örneğin blok şeması

Şekil 4 de örneğin devresi verilmiştir. PCF8583 gerçek zaman entegresi mikrokontrolörün RC3 ve RC4 bacaklarına bağlanmıştır (mikro BASIC derleyicinin I<sup>2</sup>C fonksiyonlarını kullanmak için I<sup>2</sup>C uyumlu cihazın RC3 ve RC4 bacaklarına Şekil 4 de gösterildiği gibi bağlanmalıdır). RS232 seviye dönüştürücü MAX232 entegre devresi ise mikrokontrolörün RC6 ve RC7 bacaklarına bağlanmıştır. MAX232 aynı zamanda 9 bacaklı bir konvektör ile PC mizin seri portuna takılmıştır. LCD ise mikrokontrolörün PORT B bacaklarına Şekil 4 de gösterildiği gibi bağlanmıştır. Devrede PIC18F452 modeli ve 8MHz kristal ile çalışan mikrokontrolör kullanılmıştır.

Örneğin yazılımı Şekil 5 de verilmiştir. Programın çalışması şu şekildedir:

- LCD yi konfigür yap (LCD PORT B bacaklarına bağlanmıştır)
- USART 1 konfigür yap (USART RC6 ve RC7 bacaklarına bağlanmıştır)
- I2C yi konfigür yap (I2C bus, RC3 ve RC4 bacaklarını kullanmaktadır)

- LCD ye mesaj gönder
- Tarih ve Saati PC klavyesinden oku
- Tarih ve saati PCF8583 entegreye yükle
- Tarih ve saati PCF8583 entegreden oku
- Tarih ve saati LCD de göster
- 1 saniye bekle
- Yukarıdaki 4 işlemi tekrarla

Programda şu fonksiyonlar ve alt-programlar kullanılmıştır:

**Yeni\_Satir:** Bu alt-program PC ye yeni satır karakterleri göndermektedir. Yeni satır için PC ye şu karakterler gönderilir: 0x0A ve 0x0D.

**Saati\_PCden\_Oku:** Bu alt-program PC ekranına mesaj gönderip tarih ve saatin klavyeden girilmesini bekler. PC ekranına şu mesaj gönderilmektedir:

**Tarih ve Saati Giriniz:**

Tarih ve saat şu formatta girilip ENTER tuşu tıklanmalıdır (aşağıdaki örnekte tarihin 12 Eylül 2008, ve saatin 10 u 5 dakika ve 2 saniye geçiyor olduğu kabul edilmiştir):

12/09/2008 10:05:02

**Saati\_Göster:** Bu alt-program **Usart** dizisine daha önce yüklenmiş olan tarih ve saat bilgilerini LCD de gösterir. Tarih LCD nin ilk satırında ve saat ise ikinci satırında gösterilmektedir.

**MSBB:** Bu fonksiyon verilen iki rakamlı bir BCD sayının sol rakamını ASCII ye dönüştürür. Örnek olarak, x değişkeninin ikili "0010 0101" değeri olduğunu göz önünde bulundurursak,

**MSBB (x)**

fonksiyonu ASCII "2" karakterini verir.

**LSBB:** Bu fonksiyon verilen iki rakamlı bir BCD sayının sağdaki rakamını ASCII ye dönüştürür. Örnek olarak, x değişkeninin ikili "0010 0101" değeri olduğunu göz önünde bulundurursak,

**LSBB(x)**

fonksiyonu ASCII "5" karakterini verir

**CLKD:** Bu fonksiyon verilen iki rakamlı bir ASCII sayıyı BCD ye dönüştürür. Örnek olarak x değişkeni "2" ve y değişkeninin "5" değerlerde olduğunu göz önünde bulundurursak,

**CLKD(x, y)**

fonksiyonu 25 sayısını verir (BCD "0010 0101")

**PCF8583\_Yaz:** Bu alt-program I2C fonksiyonlarını kullanarak PCF8583 entegresine gerçek zaman tarih ve saati yükler.

**PCF8583\_Oku:** Bu alt-program I2C fonksiyonlarını kullanarak PCF8583 entegreden gerçek zaman tarih ve saat verilerini okur. Okunan verile **Usart** isimli bir dizide saklanır

PCF8583 entegresine yazıp okurken yıl sadece 2 bitlik bir yazmaçta saklanmaktadır. Bu dururumda yıl şu değerleri alabilmektedir: 0, 1, 2 ve 3. Her sene yılbaşında bu yıl yazmacındaki sayı otomatik olarak entegre tarafından bir artırılmaktadır. 2 bitlik bu yıl yazmacını kullanarak bulmuş olduğumuz yılı şu şekilde hesaplayabiliriz: İlk olarak yılın soldaki 2 rakamının "20" olduğunu kabul edelim. PC den tarih ve saati okuduktan sonra başlangıç yılının sağdaki iki rakamını (örneğin 2008 yılı için 8 sayısı) mikrokontrolörün 0 adresli EEPROM belleğinde saklayabiliriz. PCF8583 entegreden yılı okuduktan sonra, gerçek yılı bulmak için okumuş olduğumuz sayıya EEPROM un 0 adresindeki sayıyı ilave edebiliriz. Örneğin, PC klavyesinden yılın 2008 olarak girildiğini farz edelim. Bu durumda EEPROM un 0 nolu adresinde 8 sayısını ve yıl yazmacının 2 bitinde ise 0 sayısını saklarız. Şimdi yılın 2009 olduğunu kabul edelim. Bu durumda yıl yazmacındaki sayı otomatik olarak 1 olacaktır. Bulunmuş olduğumuz yılı hesaplamak için EEPROM un 0 adresindeki sayıyı okuyup bu sayıya yıl yazmacındaki sayıyı ilave edersek, ve yılın ilk iki rakamının "20" olduğunu kabul edersek, sağdaki iki rakam  $8 + 1 = 9$  yani 2009 yılını elde etmiş oluruz. Bu şekilde programımız 4 yıl için sorunsuz olarak çalışacaktır. Eğer tarihi yeniden yüklemeyen 4 yıldan fazla kullanacaksa yıl için başka bir metot uygulamamız gerekecektir.

PCF8583 entegresine yazıp okumak için I<sup>2</sup>C fonksiyonlarını kullanırız. Örneğin, entegreden okumak için şu işlemleri yaparız:

- I<sup>2</sup>C bus ı başlatınız (I2C\_START)
- Cihazın adresini gönderiniz (burada 0xA0)
- Saniye yazmacını adresleyiniz (burada 2)
- I2C\_Rd fonksiyonu ile saniyeyi okuyunuz
- I2C\_Rd fonksiyonu ile dakikayı okuyunuz
- I2C\_Rd fonksiyonu ile saati okuyunuz
- I2C\_Rd fonksiyonu ile yıl/gün okuyunuz
- Yılı ayarlayınız
- Hafta gününü/ayı okuyunuz
- I<sup>2</sup>C bus ı durdurunuz (I2C\_STOP)

PC ile iletişim kurmak için mikrokontrolörü PC nin COM1 veya COM2 seri portuna bağlayınız. PC de HyperTerminal iletişim programını çalıştırınız, seri portu seçiniz ve iletişim hızını 2400 baud olarak konfigür yapınız.


## 4. SONUÇ

Gerçek zaman entegreleri mutlak zamanla ilgili olan her çeşit projede kullanılmaktadır. Örneğin, veri toplama işlemlerinde verinin toplandığı saat ve tarihi de bilmek isteriz. Bunun gibi birçok projelerde gerçek zaman entegresi kullanılmaktadır. Genel olarak bu entegrelerin çalışmaları güç kaynağı olmadığı zamanlarda pil ile desteklenmektedir.

Bu yazımızda PCF8583 gerçek zaman entegresinin nasıl kullanıldığına bir göz attık ve bu entegrenin PIC mikrokontrolör destekli projelerde nasıl kullanıldığını bir örnekle gördük.

## 5. KAYNAKÇA

1. PCF8583 User Guide  
<http://www.philipps.com>
2. mikroBASIC User Guide  
<http://www.mikroe.com>
3. PIC18F452 Data Sheet  
<http://www.microchip.com>



Şekil 4 Projenin elektronik devresi

\*\*\*\*\*

GERCEK ZAMAN SAAT

=====

' Bu projede PIC18F452 modeli bir mikrokontrolör kullanılmıştır.  
 ' Mikrokontrolörün RC3 ve RC4 bacakları PCF8583 modeli gerçek zaman entegresine bağlanmıştır.  
 ' Mikrokontrolörün RC6 ve RC7 bacakları MAX22 modeli bir RS232 seviye donusturucu entegre devreye ve oradan da 9 bacaklı RS232 konektörüne bağlanmıştır.  
 ' Mikrokontrolörün PORT B bacakları 2 satırlı bir LCD ye bağlanmıştır.  
 ' PCF8684 entegresi PC den gelen komutlarla gerçek zaman tarih ve zamana göre ayarlanmaktadır. Daha sonra her saniye gerçek zaman LCD de su formatta gösterilmektedir.  
 ' 10:00:00  
 ' 12/09/2008

\*\*\*\*\*

program ZAMAN

Dim Usart as string[19]  
 Dim Usart\_tarih as string[10]

```

Dim Usart_saat as string[8]
Dim Mesaj as string[24]

'
' Bu fonksiyon PC ye yeni satir karakterleri gonderir
'
sub procedure Yeni_Satir()
    Usart_Write(0x0A)
    Usart_Write(0x0D)
end sub

'
' Bu fonksiyon saati PC den okur
'
sub procedure Saati_PCden_Oku
    Dim k,bayrak as Byte
    Yeni_Satir()

    k = 0
    While Mesaj[k] <> 0x0
        Usart_Write(Mesaj[k])
        k = k+1
    wend

    k = 0
    bayrak = 1
    While bayrak = 1
        if Usart_Data_Ready() Then
            Usart[k] = Usart_Read()
            Usart_Write(Usart[k])
            If Usart[k] = 0x0D Then
                bayrak = 0
            End if
            k = k+1
        End if
    Wend
end sub

'
' Tarih ve saati LCD goster
'
sub procedure Saati_Goster()
    Dim j as byte

    Lcd_Cmd(LCD_CLEAR)
    FOR j = 0 To 9
        Usart_tarih[j] = Usart[j]
    Next j

    For j = 11 To 18
        Usart_saat[j-11] = Usart[j]
    Next j

    Lcd_Cmd(LCD_CLEAR)
    Lcd_Out(1,1,Usart_tarih)
    Lcd_Out(2,1, Usart_saat)
end sub

```

' Bu fonksiyon verilen ikili bir BCD sayinin sol rakamini ASCII ye donusturur

```
sub function MSBB(Dim x as Byte) as Byte
    result = ((x >> 4) + "0")
end sub
```

' Bu fonksiyon verilen ikili bir BCD sayinin sag rakamini ASCII ye donusturur

```
sub function LSBB(Dim x as Byte) as Byte
    result = ((x AND 0x0F) + "0")
end sub
```

' Bu fonksiyon verilen iki rakamli bir ASCII sayiyi BCD ye donusturur

```
sub function CLKD(Dim x,y as Byte)as Byte
    result = (x - "0") << 4
    result = result + y - "0"
end sub
```

' Bu fonksiyon PC den okunan tarih ve saati PCF8583 entegreye yukler  
' (tarih ve saat Usart isimli dizide saklidir)

```
sub procedure PCF8583_Yaz()
    dim sene as Byte
```

```
    I2C_Start()                ' I2C baslat
    I2C_Wr(0xA0)                ' PCF8583 adresle
    I2C_Wr(0)
    I2C_Wr(0x80)
    I2C_Wr(0)                    ' Saniyeleri yaz
    I2C_Wr(CLKD(Usart[17],Usart[18])) ' Dakikalari yaz
    I2C_Wr(CLKD(Usart[14],Usart[15])) ' Saati yaz
    I2C_Wr(CLKD(Usart[11],Usart[12]))
    sene = 10*(Usart[8] - "0")+Usart[9] - "0"
    Eeprom_Write(0,sene)        ' Senenin sag iki rakami
    I2C_Wr(CLKD(Usart[0], Usart[1])) ' Gunu yaz
    I2C_Wr(CLKD(Usart[3],Usart[4])) ' Ayi yaz
    I2C_Stop()                  ' I2C durdur
```

```
    I2C_Start()                ' I2C baslat
    I2C_Wr(0xA0)                ' PCF8583 adresle
    I2C_Wr(0)                    ' Sayaci baslat
    I2C_Wr(0)
    I2C_Stop()                  ' I2C durdur
```

```
end sub
```

' Bu fonksiyon tarih ve saati PCF8583 entegreden okur  
' ve Usart dizisinde saklar

```
sub procedure PCF8583_Oku()
    Dim rtc,sene,yuksek,dusuk as Byte
```

```
    I2C_Start()                ' I2C baslat
    I2C_Wr(0xA0)                ' PCF8583 adresi
    I2C_Wr(2)
```

```

I2C_Repeated_Start()
I2C_Wr(0xA1)           ' Okumak icin adres
rtc = I2C_Rd(1)        ' Saniyeleri oku
Usart[17] = MSBB(rtc)
Usart[18] = LSBB(rtc)
rtc = I2C_Rd(1)        ' Dakikalari oku
Usart[14] = MSBB(rtc)
Usart[15] = LSBB(rtc)
rtc = I2C_Rd(1)        ' Saati oku
Usart[11] = MSBB(rtc)
Usart[12] = LSBB(rtc)
rtc = I2C_Rd(1)        ' sene/gun oku
Usart[0] = MSBB((rtc AND 0x30))
Usart[1] = LSBB(rtc)
sene = (rtc AND 0xc0) >> 6
rtc = Eeprom_Read(0)   ' Senenin sag iki rakami
sene = sene + rtc      ' Seneyi ayarla
yukse = sene / 10     ' Ust bayti
dusuk = sene MOD 10   ' Alt bayti
Usart[8] = yuksek + "0" ' ASCII ye donustur
Usart[9] = dusuk + "0" ' ASCII ye donustur
Usart[6] = "2"        ' Senenin ilk bayti
Usart[7] = "0"        ' Senenin ikinci bayti
rtc = I2C_Rd(0)       ' AY i oku
Usart[3] = MSBB((rtc and 0x10))
Usart[4] = LSBB(rtc)
I2C_Stop()           ' I2C yi durdur
end sub

```

```

,
' ***** ANA PROGRAMIN BASLANGICI *****
,

```

```
main:
```

```

    TRISB = 0
    Usart = "dd/mm/yyyy hh mm ss"
    Usart_tarih = "dd/mm/yyyy"
    Usart_saat = "hh:mm:ss"
    Mesaj = "Tarih ve Saati Giriniz: "

```

```

' LCD yi baslat
,

```

```
    LCD_Config(PORTB,3,2,1,0,PORTB,4,7,5)
```

```

' USART i baslat
,

```

```
    Usart_Init(2400)
```

```

' LCD ye baslangic mesaji gonder
,

```

```
    Lcd_Out(1,1,"PCF8583")
```

```

' I2C Bus i baslat
,

```

```
    I2C_Init(100000)
```

```

' Saati seri portdan oku
,

```

```
    Saati_PCden_Oku()
```

```
'  
' Tarih ve saati PCF8583 entegreye yukle  
'  
    PCF8583_Yaz()  
'  
' Saat ve tarihi LCD de her saniye goster  
'  
    While True  
        PCF8583_Oku()  
        Saati_Goster()  
        Delay_Ms(1000)  
    Wend
```

End.

**Şekil 5** Projenin yazılımı